Maple Lecture 2. Getting Started and Getting Help

In this lecture we show how to use Maple as a calculator and explore the extensive help facilities. The material in this note corresponds to the beginning of the second chapter of [1]. See also [2, Chapter 1].

2.1 Getting Started

At a basic level, we type in a command and Maple displays the result. Every command you type in must be terminated by a semicolon (`;`) or a colon (`:`). In case of a semicolon, the result is displayed, with a colon, the display of the result is suppressed. Please note that Maple is case sensitive.

```
[> 34^34;                # Maple can handle very large numbers
```

To type in large numbers: the continuation symbol \ may be handy:

```
[> 23324\
[> 23455;
```

to group the digits of long numbers.

We can recall the result of the last operation with the ditto operator `%`. Here we will use this to assign the value of the last operation to a variable. In doing so with the ditto operator `%`, we do not recalculate anything, which is efficient in case of lengthy procedures.

```
[> b := %;
```

Note that := stands for assignment, unlike C/C++.

With `%%` we get the next to last result, and to get the result before that, we do

```
[> a := %%%;
```

To recall the values, we type for the variables `a` and `b`:

```
[> a; b;
```

We can clear all variables by the restart command (there is also the restart button):

```
[> restart;
[> a; b;
```

It is a good practice to start every worksheet with the `restart` command, because all variables are shared among all worksheets. If a `restart` occurs at the beginning of a worksheet, then the **Execute Worksheet** option from the **Edit** menu first clears the memory before running through all the instructions.

While Maple likes to work exactly, transcendental numbers have no finite numerical representation. Therefore we need to approximate. For example, to see the first 30 places of π, we type

```
[> Pi30 := evalf(Pi,30);
```

Nevertheless, sometimes it is better to wait with the approximations, as Maple computes with symbols. Observe what happens in the following sequence of commands:

```
[> sin(Pi);              # evaluate sine function at Pi
[> sin(Pi30);            # evaluate at approximation of Pi
[> % - %%;               # see the difference
```

2.2 Getting Help

Maple can be quite overwhelming to the novice. Navigating through the help browser shows the structure of the software system.

Maple has an extensive help facility. There is the Help menu at the upper right corner. We may also launch the Help facility from the prompt:

```
[> ?help;                          # launches the browser
```

With the path

```
  Mathematics... -> Algebra... -> Expression Manipulation... -> Factoring... -> factor
```

we get down to the leaf page with explanation on how to factor a multivariate polynomial. Observe the navigation arrows from the toolbar.

We can also get direct information about the factor command:

```
[> ?factor;                        # shows the description of the command
```

or alternatively we type `info(factor);` at the prompt. Information about a command contains a specification of the command, some examples, and related issues. These topics can be requested separatedly:

```
[> usage(factor);                  # shows the specification, syntax
[> example(factor);                # examples
[> related(factor);                # related topics
```

Another powerful tool to get access to the help system is via the **Topic Search...** option of the **Help** menu. Launch this option and type in `f` in the `Topic:` field and you will see all commands starting with `f`. Selecting from the matching topics leads to the help page. Typing more letters narrows the search.

2.3 The Maple Library

Maple offers a wealth of mathematical knowledge in its standard libary, through various packages and the share library with user contributions. For an overview:

```
[> ?index,packages;
```

To use packages invoke the **with** command. Instead of loading an entire package (which may overwrite names you want to use), we can load only one function, e.g.:

```
[> with(orthopoly,T);              # use Chebychev orthogonal polynomials
[> T(4,x);                         # 4-th Chebychev orthogonal polynomial in x
```

Instead of loading at all (if we want to use the symbol `T` for something else), we may call it with its full name:

```
[> orthopoly[T](4,y);              # 4-th Chebychev orthogonal polynomial in y
```

To avoid tedious typing of long names, we can work with an **alias**:

```
[> alias(Cheby = orthopoly[T]);  # give other name to orthopoly[T]
[> Cheby(4,z);                   # 4-th Chebychev orthogonal polynomial in z
```

2.4 Assignments

1. Consider the following Maple session:

```
[> 4^2;
                              16
[> 2^3:
[> % + %%;
                              24
```

   Explain how the number `24` as result of the last instruction was obtained. What happens if you execute the last instruction over and over again?

2. Explain the different results of the following Maple commands: (a) `x:y;` (b) `x/y;` (c) `x\y;`

3. Explain the difference between `Pi` and `pi` in Maple. Give Maple commands which reveal the difference between `Pi` and `pi`.

4. What is the difference between `e` and **`e`** in Maple?
   (*Hint*: try to evaluate the transcendental number e to 30 decimal places.)
   Show how you can "reserve" the letter `e` for the transcendental number e.

5. Execute the commands **evalf(tan(Pi/2))** and **tan(evalf(Pi/2))** and interpret the results. Why do these two commands give different answers? Which of the two commands gives the correct answer? Also compare **evalf(tan(Pi))** with **tan(evalf(Pi))**.

6. A shortcut key provides fast access to the capabilities offered in the menus. How do you find an overview in the help system of all shortcut keys for Windows? Give the path in the help browser which leads to the leaf with the Shortcut Key Summary for Windows.

7. Give the Maple command(s) to show the help page on converting algebraic expressions to a continued-fraction form. Write down the highlighted subject titles you see in the 5-column area at the top Maple's Help Browser when viewing this help page.

8. The following assignments are chosen to develop your abilities in using Maple's help system:

   (a) Given an equation, like $\ln(10^x) = x\ln(10)$, how do you select the left or the right hand side of it?

   (b) A continued fraction approximation of e^x of order four is

   $$1 + \frac{x}{1 + \frac{x}{-2 - \frac{x}{3}}}.$$

   Find the Maple command to find this result.

   (c) The polynomial $p(x) = 2x^5 + x + 4$ factors over $\mathbb{Z}_7 = \{0, 1, 2, 3, 4, 5, 6\}$.
   Find the factorization of $p(x)$ over $\mathbb{Z}_7$.

   (d) Let $S = \{1, 2, 3, 4, 5\}$, find 2^S, i.e.: all subsets of S.

9. The command **stats[random,uniform[0,1]](20)** returns 20 random numbers, uniformly distibuted between 0 and 1. Since this command is so long, we wish to create an abbreviation: **ud**.

   Give the Maple command to create **ud**, so that **ud(20)** is equivalent to **stats[random,uniform[0,1]](20)**.

References

[1] A. Heck. *Introduction to Maple*. Springer-Verlag, third edition, 2003.

[2] D.I. Schwartz. *Introduction to Maple 8*. Prentice-Hall, 2003.